Kapitel 2.32 : Lösungen der Übungsaufgaben 

83. Schauen Sie sich diese Funktion an. Finden Sie heraus, was Sie tut, kritisieren Sie sie, und schreiben Sie dann eine neue Version, die den gleichen Zweck erfüllt (lassen Sie aber das weg, was nicht hineingehört).


(defun abcde(zzz / liste elem pkt1 pkt2 pliste)

  (setq pliste nil)

  (setq elem zzz)

  (command "_UCS" "_W")

  (setq pkt1 (cdr (assoc 10 elem)))

  (setq pkt2 (cdr (assoc 11 elem)))

  (setq pliste (append pliste (list pkt1)))
  (setq pliste (append pliste (list pkt2)))

  (setq ppliste (append ppliste (list pliste))

)

Die Funktion packt die Inhalte der Elementgruppencodes 10 und 11 in eine Liste und gibt diese Liste zurück.

· Die Funktion hat keinen selbsterklärenden Namen.

· Der übergebene Parameter (Variable zzz) ebensowenig.
· Es steht eine unbenutzte Variable liste bei den lokalen Variablen.
· (setq pliste nil) - Variablen werden in AutoLISP nicht initialisiert, sie sind nil, wenn sie nicht an einen Wert gebunden sind.
· (setq elem zzz) – ist nicht notwendig, weil genauso gut mit zzz weitergearbeitet werden kann. Also kann elem somit auch aus der Liste der lokalen Variablen entfernt werden.
· (command "_UCS" "_W") – völliger Unsinn, weil direkt mit den Elementdaten eines Objekts gearbeitet wird.
· (append pliste (list pkt1)) – pkt1 ist bereits eine Liste. Hier wird recht umständlich ein neues Listenelement in eine bestehende Liste mit append eingefügt.
· (setq ppliste (append ppliste (list pliste)) – Der Gipfel des Unsinns. pliste ist längst eine Liste. Es wird lediglich eine zusätzliche Verschachtelungstiefe erzeugt. Erschwerend kommt hinzu, dass plötzlich eine neue Variable ppliste eingesetzt wird. Sie ist nicht als lokal definiert, was bedeutet, dass die Rückgabe immer länger wird, je öfter Sie diese Funktion laufen lassen, weil alle vorherigen Ergebnisse mitgeschleppt werden. Es könnte sich bei ppliste allerdings auch um einen Tippfehler handeln.
Hier eine Variante, wie diese Funktion bereinigt aussehen könnte:

(defun LinienEndpunkte (elementdaten /)

  (list

    (cdr (assoc 10 elementdaten))

    (cdr (assoc 11 elementdaten))

  )

)

Alternativ könnte der Funktionsteil auch so aussehen:

(cons

  (cdr (assoc 10 elementdaten))

  (list (cdr (assoc 11 elementdaten)))

)

oder aber so:

(mapcar

  '(lambda (arg)

     (cdr (assoc arg elementdaten))

   )

  '(10 11)

)

